NAG Library Routine Document

F03ABF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F03ABF calculates the determinant of a real symmetric positive definite matrix using a Cholesky factorization.

2 Specification

```
SUBROUTINE F03ABF (A, LDA, N, DET, WKSPCE, IFAIL)

INTEGER            LDA, N, IFAIL
REAL (KIND=nag_wp) A(LDA,*), DET, WKSPCE(max(1,N))
```

3 Description

The determinant of A is calculated using the Cholesky factorization $A = LL^{\mathrm{T}}$, where L is lower triangular. The determinant of A is the product of the squares of the diagonal elements of L.

4 References

Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer–Verlag

5 Parameters

1: A(LDA,$*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the upper triangle of the n by n positive definite symmetric matrix A. The elements of the array below the diagonal need not be set.

On exit: the subdiagonal elements of the lower triangular matrix L. The upper triangle of A is unchanged.

2: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F03ABF is called.

Constraint: LDA $\geq \max(1, \mathrm{N})$.

3: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

4: DET – REAL (KIND=nag_wp) *Output*

On exit: the determinant of A.

5: WKSPCE(max(1, N)) – REAL (KIND=nag_wp) array *Workspace*

6: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value −1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value −1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

The matrix A is not positive definite, possibly due to rounding errors. The factorization could not be completed. DET is set to 0.0.

IFAIL = 2

Overflow. The value of the determinant is too large to be held in the computer.

IFAIL = 3

Underflow. The value of the determinant is too small to be held in the computer.

IFAIL = 4

On entry, N < 0,
or LDA < max(1, N).

7 Accuracy

The accuracy of the determinant depends on the conditioning of the original matrix. For a detailed error analysis see page 25 of Wilkinson and Reinsch (1971).

8 Further Comments

The time taken by F03ABF is approximately proportional to n^3.

9 Example

This example calculates the determinant of the real symmetric positive definite matrix

$$\begin{pmatrix} 5 & 7 & 6 & 5 \\ 7 & 10 & 8 & 7 \\ 6 & 8 & 10 & 9 \\ 5 & 7 & 9 & 10 \end{pmatrix}.$$

9.1 Program Text

```
    Program f03abfe

!     F03ABF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: f03abf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: det
      Integer                          :: i, ifail, lda, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: a(:,:), wkspce(:)
!     .. Executable Statements ..
      Write (nout,*) 'F03ABF Example Program Results'

!     Skip heading in data file
      Read (nin,*)

      Read (nin,*) n
      lda = n
      Allocate (a(lda,n),wkspce(n))

      Read (nin,*)(a(i,1:n),i=1,n)

      ifail = 0
      Call f03abf(a,lda,n,det,wkspce,ifail)

      Write (nout,*)
      Write (nout,99999) 'Value of determinant = ', det

99999 Format (1X,A,F9.4)
    End Program f03abfe
```

9.2 Program Data

```
F03ABF Example Program Data
  4
    5    7    6    5
    7   10    8    7
    6    8   10    9
    5    7    9   10
```

9.3 Program Results

```
 F03ABF Example Program Results

 Value of determinant =    1.0000
```